

Practical VBA code

 Place the code below in a new module and call it modProgressMeter

```
Option Compare Database
Option Explicit
```

```
Public Function ProgressMeter(rst_qryName As String)

    Dim db As DAO.Database
    Dim rst As DAO.Recordset
    Dim lngCount As Long
    Dim intProgressAmount As Integer
    Dim strProgressMsg As String
    Dim varRetVal As Variant

    Set db = CurrentDb()
    Set rst = db.OpenRecordset(rst_qryName)

    ' Assure that recordset has at least 1 record
    ' before running the progress meter else clean
    ' up and close recordset.

    If rst.RecordCount > 0 Then

        ' Move to the last record of query
        ' to determine total count of records

        rst.MoveLast
        lngCount = rst.RecordCount

        ' Move back to the first record of query

        rst.MoveFirst

        ' Initialize the progress meter

        strProgressMsg = "Your Text Here . . ."
        varRetVal = SysCmd(acSysCmdInitMeter, strProgressMsg, lngCount)

        ' Enumerate through all the records

        For intProgressAmount = 1 To lngCount
            varRetVal = SysCmd(acSysCmdUpdateMeter, intProgressAmount)
            DoEvents
            rst.MoveNext
        Next intProgressAmount

        ' Remove the progress meter

        varRetVal = SysCmd(acSysCmdRemoveMeter)

        ' Clean up by closing recordsets
        ' and then setting them equal to nothing

        rst.Close
        Set rst = Nothing
        Set db = Nothing
    Else
        rst.Close
        Set rst = Nothing
        Set db = Nothing
    End If

End Function
```

 Place the code below in any or every bound form you wish to see the Progress Meter

```
Private Sub Form_Open(Cancel As Integer)

    ' Show Progress Meter
    Call ProgressMeter("Your_Query_Name_Here")

End Sub
```